NAG Library Routine Document

F08LEF (DGBBRD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08LEF (DGBBRD) reduces a real m by n band matrix to upper bidiagonal form.

2 Specification

```
SUBROUTINE F08LEF (VECT, M, N, NCC, KL, KU, AB, LDAB, D, E, Q, LDQ, PT,        &
                   LDPT, C, LDC, WORK, INFO)

INTEGER            M, N, NCC, KL, KU, LDAB, LDQ, LDPT, LDC, INFO
REAL (KIND=nag_wp) AB(LDAB,*), D(min(M,N)), E(min(M,N)-1), Q(LDQ,*),           &
                   PT(LDPT,*), C(LDC,*), WORK(2*max(M,N))
CHARACTER(1)       VECT
```

The routine may be called by its LAPACK name ***dgbbrd***.

3 Description

F08LEF (DGBBRD) reduces a real m by n band matrix to upper bidiagonal form B by an orthogonal transformation: $A = QBP^{\mathrm{T}}$. The orthogonal matrices Q and P^{T}, of order m and n respectively, are determined as a product of Givens rotation matrices, and may be formed explicitly by the routine if required. A matrix C may also be updated to give $\tilde{C} = Q^{\mathrm{T}}C$.

The routine uses a vectorizable form of the reduction.

4 References

None.

5 Parameters

1: VECT – CHARACTER(1) *Input*

On entry: indicates whether the matrices Q and/or P^{T} are generated.

VECT = 'N'
: Neither Q nor P^{T} is generated.

VECT = 'Q'
: Q is generated.

VECT = 'P'
: P^{T} is generated.

VECT = 'B'
: Both Q and P^{T} are generated.

Constraint: VECT = 'N', 'Q', 'P' or 'B'.

2: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

3: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

4: NCC – INTEGER *Input*

On entry: n_C, the number of columns of the matrix C.

Constraint: $\mathrm{NCC} \geq 0$.

5: KL – INTEGER *Input*

On entry: the number of subdiagonals, k_l, within the band of A.

Constraint: $\mathrm{KL} \geq 0$.

6: KU – INTEGER *Input*

On entry: the number of superdiagonals, k_u, within the band of A.

Constraint: $\mathrm{KU} \geq 0$.

7: AB(LDAB,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1,\mathrm{N})$.

On entry: the original m by n band matrix A.

The matrix is stored in rows 1 to k_l+k_u+1, more precisely, the element A_{ij} must be stored in

$$\mathrm{AB}(k_u+1+i-j,j) \qquad \text{for } \max(1,j-k_u) \leq i \leq \min(m,j+k_l).$$

On exit: AB is overwritten by values generated during the reduction.

8: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08LEF (DGBBRD) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KL}+\mathrm{KU}+1$.

9: D($\min(\mathrm{M},\mathrm{N})$) – REAL (KIND=nag_wp) array *Output*

On exit: the diagonal elements of the bidiagonal matrix B.

10: E($\min(\mathrm{M},\mathrm{N})-1$) – REAL (KIND=nag_wp) array *Output*

On exit: the superdiagonal elements of the bidiagonal matrix B.

11: Q(LDQ,$*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array Q must be at least $\max(1,\mathrm{M})$ if VECT = 'Q' or 'B', and at least 1 otherwise.

On exit: if VECT = 'Q' or 'B', contains the m by m orthogonal matrix Q.

If VECT = 'N' or 'P', Q is not referenced.

12: LDQ – INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F08LEF (DGBBRD) is called.

Constraints:

if VECT = 'Q' or 'B', $\mathrm{LDQ} \geq \max(1,\mathrm{M})$;
otherwise $\mathrm{LDQ} \geq 1$.

13: PT(LDPT,$*$) – REAL (KIND=nag_wp) array *Output*

Note: the second dimension of the array PT must be at least $\max(1, N)$ if VECT = 'P' or 'B', and at least 1 otherwise.

On exit: the n by n orthogonal matrix P^{T}, if VECT = 'P' or 'B'. If VECT = 'N' or 'Q', PT is not referenced.

14: LDPT – INTEGER *Input*

On entry: the first dimension of the array PT as declared in the (sub)program from which F08LEF (DGBBRD) is called.

Constraints:

if VECT = 'P' or 'B', LDPT $\geq \max(1, N)$;
otherwise LDPT ≥ 1.

15: C(LDC,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array C must be at least $\max(1, NCC)$.

On entry: an m by n_C matrix C.

On exit: C is overwritten by $Q^{T}C$. If NCC $= 0$, C is not referenced.

16: LDC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which F08LEF (DGBBRD) is called.

Constraints:

if NCC > 0, LDC $\geq \max(1, M)$;
if NCC $= 0$, LDC ≥ 1.

17: WORK($2 \times \max(M, N)$) – REAL (KIND=nag_wp) array *Workspace*

18: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed bidiagonal form B satisfies $QBP^{T} = A + E$, where

$$\|E\|_2 \leq c(n)\epsilon\|A\|_2,$$

$c(n)$ is a modestly increasing function of n, and ϵ is the ***machine precision***.

The elements of B themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the singular values and vectors.

The computed matrix Q differs from an exactly orthogonal matrix by a matrix F such that

$$\|F\|_2 = O(\epsilon).$$

A similar statement holds for the computed matrix P^{T}.

8 Further Comments

The total number of real floating point operations is approximately the sum of:

$6n^2k$, if VECT = 'N' and NCC = 0, and

$3n^2n_C(k-1)/k$, if C is updated, and

$3n^3(k-1)/k$, if either Q or P^{T} is generated (double this if both),

where $k = k_l + k_u$, assuming $n \gg k$. For this section we assume that $m = n$.

The complex analogue of this routine is F08LSF (ZGBBRD).

9 Example

This example reduces the matrix A to upper bidiagonal form, where

$$A = \begin{pmatrix} -0.57 & -1.28 & 0.00 & 0.00 \\ -1.93 & 1.08 & -0.31 & 0.00 \\ 2.30 & 0.24 & 0.40 & -0.35 \\ 0.00 & 0.64 & -0.66 & 0.08 \\ 0.00 & 0.00 & 0.15 & -2.13 \\ -0.00 & 0.00 & 0.00 & 0.50 \end{pmatrix}.$$

9.1 Program Text

```
    Program f08lefe

!     F08LEF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: dgbbrd, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
      Character (1), Parameter         :: vect = 'B'
!     .. Local Scalars ..
      Integer                          :: i, info, j, kl, ku, ldab, ldb, ldc,  &
                                          ldpt, ldq, m, n, ncc
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:), b(:,:), c(:,:), d(:), e(:), &
                                          pt(:,:), q(:,:), work(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: abs, max, min
!     .. Executable Statements ..
      Continue
      Write (nout,*) 'F08LEF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n, kl, ku, ncc
      ldab = kl + ku + 1
      ldb = m
      ldc = m
      ldpt = n
      ldq = m
      Allocate (ab(ldab,n),b(ldb,n),c(m,ncc),d(n),e(n-1),pt(ldpt,n),q(ldq,m), &
        work(2*m+2*n))

!     Read A from data file

      Read (nin,*)((ab(ku+1+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,m)

!     Reduce A to upper bidiagonal form
!     The NAG name equivalent of dgbbrd is f08lef
```

```
      Call dgbbrd(vect,m,n,ncc,kl,ku,ab,ldab,d,e,q,ldq,pt,ldpt,c,ldc,work, &
        info)

!     Print the absolute values of bidiagonal vectors d and e.
!     Any of these can differ by a sign change by combinations of sign
!     changes in columns of Q and P (rows of PT).
      Write (nout,*)
      Write (nout,*) 'Diagonal D:'
      Write (nout,99999) abs(d(1:n))
      Write (nout,*)
      Write (nout,*) 'Off-diagonal E:'
      Write (nout,99999) abs(e(1:n-1))
99999 Format (1X,4(3X,F11.4))

    End Program f08lefe
```

9.2 Program Data

```
F08LEF Example Program Data
  6  4  2  1  0                :Values of M, N, KL, KU and NCC
 -0.57  -1.28
 -1.93   1.08  -0.31
  2.30   0.24   0.40  -0.35
         0.64  -0.66   0.08
                0.15  -2.13
                       0.50   :End of matrix A
```

9.3 Program Results

```
 F08LEF Example Program Results

 Diagonal D:
         3.0561        1.5259        0.9690        1.5685

 Off-diagonal E:
         0.6206        1.2353        1.1240
```